

eCH-0117 Méta-attributs pour modèles INTERLIS

Titre	Méta-attributs pour modèles INTERLIS
Code	eCH-0117
Type	norme de procédure
Stade	expérimentale
Version	1.00
Statut	Approuvé
Validation	2011-06-08
Date de publication	2011-06-22
Remplace	
Langues	Français, Allemand
Auteur(s)	Groupe spécialisé eCH INTERLIS <i>Claude Eisenhut, Eisenhut Informatik AG</i> <i>Michael Germann, infoGrips GmbH</i> <i>Dr. Peter Staub, swisstopo</i>
Éditeur / Distributeur	Association eCH, Mainaustrasse 30, Case postale, 8034 Zürich T 044 388 74 64, F 044 388 71 80 www.ech.ch / info@ech.ch

Condensé

Cette norme décrit la définition des méta-attributs en tant que partie d'un modèle de données INTERLIS. Les règles définies dans la présente spécification permettent d'intégrer les méta-informations dans un modèle de données INTERLIS, sans qu'il soit nécessaire d'adapter la spécification de langue [1] INTERLIS.

Les règles de syntaxe pour les méta-attributs seront adaptées dans le livre de normes d'INTERLIS. On définira notamment à quels emplacements les nouveaux méta-attributs peuvent être définis, dans la spécification INTERLIS existante.

Table des matières

1	Statut du document	3
2	Introduction	3
	2.1 Aperçu	3
	2.2 Avantages.....	3
3	Concept	3
4	Syntaxe	4
	4.1 Définitions - syntaxe	4
	4.2 Règles de syntaxe pour les méta-attributs	4
5	Positionnement au sein de la langue INTERLIS	6
6	Exemple	9
7	Exclusion de responsabilité – Droits de tiers	9
8	Droits d'auteur	10
	Annexe– Références et Bibliographie	11

1 Statut du document

Le Comité d'experts a **approuvé** le présent document, lui conférant force normative pour le domaine d'application défini et dans les limites de validité fixées.

2 Introduction

2.1 Aperçu

Les méta-attributs permettent de compléter le modèle INTERLIS par des indications supplémentaires qui ne sont pas prévues dans la spécification actuelle d'INTERLIS [1].

Dans le document relatif au méta-modèle -INTERLIS 2 [2], on trouve le texte suivant:

«Des méta-attributs compositeurs peuvent être affectés à un méta-élément (classe «Méta-attribut»). Les méta-attributs ne sont pas davantage définis, ni par la langue INTERLIS, ni par le méta-modèle. Ils servent surtout à ce que les informations sortant d'INTERLIS puissent être ajoutées en tant que partie intégrante des données de modèle. Les méta-attributs possèdent un nom (Attribut « Nom ») qui doit être unique au sein des méta-attributs du même méta-élément. Ils possèdent également la valeur descriptive (Attribut « Valeur»).»

Le présent document décrit une syntaxe permettant de formuler directement ce genre de méta-attributs en tant que partie du modèle, contrairement au méta-modèle, où l'on définit comment les méta-attributs sont définis et sont transférés sous forme de fichier XML en tant que données.

2.2 Avantages

La définition directe des méta-attributs dans le modèle de données présente l'avantage suivant: les méta-informations peuvent être immédiatement saisies pour les constructions INTERLIS correspondantes qu'elles doivent influencer. La spécification présente est conçue de telle sorte que les méta-attributs peuvent être appliqués *sans* modification de la spécification INTERLIS [1].

À la différence du commentaire de ligne normal, qui possède exclusivement un caractère informatif, les méta-attributs peuvent être interprétés par les outils de logiciel correspondants, conformément aux règles de syntaxe de la section 4.2 et être transférés en vue d'un traitement ultérieur, le cas échéant.

3 Concept

Afin de ne pas avoir à modifier la définition de langue existante, les méta-attributs sont formulés au sein des commentaires de ligne INTERLIS. Un commentaire de ligne INTERLIS est défini comme suit, conformément à [1]:

«Un commentaire de ligne s'ouvre avec deux points d'exclamation qui se suivent directement. Le commentaire de ligne se termine par une fin de ligne.»

Afin de pouvoir différencier les méta-attributs des véritables commentaires, il est impératif d'ajouter le caractère « @ » en tant que troisième caractère. Cela signifie donc qu'un commentaire de ligne contenant des méta-attributs commence par « !!@ ».

Pour savoir à quel méta-élément (conformément au méta-modèle -INTERLIS 2) un tel méta-attribut appartient, il suffit d'observer la position du commentaire de ligne. Le méta-attribut se rapporte à la première construction de langue suivante, comme par exemple une définition de modèle, de thème ou de classe.

4 Syntaxe

4.1 Définitions - syntaxe

Les règles de syntaxe sont définies selon la forme Extended Backus-Naur (EBNF), comme les autres règles de langue INTERLIS (voir [1], section 2.1).

En plus de cela, les règles de syntaxe suivantes sont également introduites:

~ A	Un caractère de tous caractères possibles mais différent de ceux définis d'«A»
'A' .. 'D'	Domaine de caractères: «A», «B», «C» et «D»
'\t'	Caractères de tabulation Unicode: <CHARACTER TABULATION>
'\f'	Caractères de saut de page Unicode: <FORM FEED (FF)>
'\r'	Caractère de retour chariot Unicode: <CARRIAGE RETURN (CR)>
'\n'	Caractère de saut de ligne Unicode: <LINE FEED (LF)>

4.2 Règles de syntaxe pour les méta-attributs

Un *Méta-attribut* se compose d'un Code préliminaire « !!@ », ainsi que d'un nom de méta-attribut et d'une valeur de méta-attribut affectée:

```
MetaAttribute = '!!@' Name '=' Value { ';' [ Name '=' Value ] } .
```

Une *Valeur de méta-attribut* se compose soit d'un nom, soit d'une chaîne libre (string).

```
Value = Metaattributename | String .
```

Un *Nom de méta-attribut* se compose d'une suite quelconque de caractères (longueur max. 255 caractères), sans caractère spécial:

```
Metaattributename = (* ~( '\t' | '\f' | '\r' | '\n' | ' ' | '=' | ';' | ',' | '"' | '\' ) *) .
```

Une *chaîne* est une suite quelconque de caractères incluant des séquences «Escape» entourées de double-guillemets:

```
String = '"' { Escape | ~( '"' | '\\' ) } '"' .
```

Une *séquence Escape* commence par un antislash suivi ou d'un double guillemet ou d'un antislash ou d'un numéro de caractère Unicode en hexadécimal à quatre chiffres:

```
Escape = '\\' ( '"' | '\\' | 'u' HexDigit HexDigit HexDigit HexDigit ) .
```

Un caractère Unicode se compose de *nombres hexadécimaux*. Ceux-ci sont constitués de chiffres et des lettres a/A à f/F:

```
HexDigit = '0' .. '9' | 'a' .. 'f' | 'A' .. 'F' .
```

5 Positionnement au sein de la langue INTERLIS

Dans les sections suivantes, on définit, à partir des règles de syntaxe du manuel de référence INTERLIS [1], à quel emplacement les méta-attributs doivent figurer. Les méta-attributs présents à d'autres emplacements sont considérés en tant que commentaires de ligne normaux et sont donc ignorés.

Définition de modèle:

```
ModelDef = { MetaAttribute } [ 'CONTRACTED' ]  
          [ 'TYPE' | 'REFSYSTEM' | 'SYMBOLGY' ] 'MODEL' Model-Name ...
```

Définition du sujet:

```
TopicDef = { MetaAttribute } [ 'VIEW' ] 'TOPIC' Topic-Name ...
```

Définition des classes:

```
ClassDef = { MetaAttribute } 'CLASS' Class-Name ...
```

Définition de la structure:

```
StructureDef = { MetaAttribute } 'STRUCTURE' Structure-Name ...
```

Définition des attributs:

```
AttributeDef = { MetaAttribute } [ [ 'CONTINUOUS' ] 'SUBDIVISION' ]  
          Attribute-Name ...
```

Définition des rôles dans les associations:

```
RoleDef = { MetaAttribute } Role-Name ...
```

Définition des domaines de valeurs:

```
DomainDef = 'DOMAIN' { { MetaAttribute } Domain-Name ...
```

Élément d'énumération:

```
EnumElement = { MetaAttribute } EnumElement-Name ...
```

Définition de forme de ligne:

```
LineFormTypeDef = 'LINE' 'FORM' { { MetaAttribute }  
  LineFormType-Name ...
```

Définition de l'unité:

```
UnitDef = 'UNIT' { { MetaAttribute } Unit-Name ...
```

Définition du contenant de méta-objet:

```
MetaDataBasketDef = { MetaAttribute } ( 'SIGN' | 'REFSYSTEM' )  
  'BASKET' Basket-Name Properties <FINAL>  
  [ 'EXTENDS' MetaDataBasketRef ] '~' TopicRef  
  { 'OBJECTS' 'OF' Class-Name ':'  
    { MetaAttribute } MetaObject-Name  
    { ',' { MetaAttribute } MetaObject-Name } } ';'.
```

Définition du paramètre:

```
ParameterDef = { MetaAttribute } Parameter-Name ...
```

Définition du paramètre de durée:

```
RunTimeParameterDef = 'PARAMETER' { { MetaAttribute }  
  RunTimeParameter-Name ...
```

Définition d'une condition de cohérence:

```
ConstraintDef = { MetaAttribute } ( MandatoryConstraint | ...
```

Définition de fonctions:

```
FunctionDef = { MetaAttribute } 'FUNCTION' Function-Name ...
```

Définition d'une vue (View):

```
ViewDef = { MetaAttribute } 'VIEW' View-Name ...
```

Définition de graphique:

```
GraphicDef = { MetaAttribute } 'GRAPHIC' Graphic-Name ...
```

Les méta-attributs d'une définition de sujet (Topic) sont dans les données de schéma, ce qui signifie que les méta-objets sont affectés au sous modèle «SubModel», conformément au méta-modèle [2] (et non à l'unité de données «DataUnit»).

Les méta-attributs d'une définition de condition de cohérence («ConstraintDef») ne sont pas possibles en tant que données de schéma, car les contraintes (Constraints) n'ont aucun identifiant OID, conformément au méta-modèles [2], et ne sont pas un « méta-élément ».

6 Exemple

L'exemple suivant doit illustrer l'utilisation des méta-attributs. Ce faisant, on partira du principe que le jeu de caractères utilisé (ISO -8859 -1) est indiqué dans un modèle de données INTERLIS. Conformément à la règle de syntaxe dans le chapitre 4, le méta-attribut correspondant se compose des éléments suivants:

Elément- méta-attribut	Fragment de code INTERLIS
Code de ligne	!!@
Nom du méta-attribut	ili.charset
Valeur du méta-attribut	ISO-8859-1

On peut également voir un exemple valable complet avec le Code-INTERLIS suivant:

```
INTERLIS 2.3;
!!@ ili.charset = ISO-8859-1
MODEL Example AT "mailto:noreply@localhost" VERSION "2010-02-30" =
END Example.
```

Le méta-attribut est affecté à un objet de la classe «Model» (modèle) à partir du méta-modèle avec le nom «Example» (exemple):

```
<IlisMeta07.ModelData BID="MODEL.Example">
  <IlisMeta07.ModelData.Model TID="Example">
    <Name>Example</Name>
    <iliVersion>2.3</iliVersion>
    <Contracted>>false</Contracted>
    <Kind>NormalM</Kind>
    <At>mailto:noreply@localhost</At>
    <Version>2010-02-30</Version>
  </IlisMeta07.ModelData.Model>
  <IlisMeta07.ModelData.MetaAttribute
    TID="Example.METAOBJECT.ili.charset">
    <Name>ili.charset</Name>
    <Value>ISO-8859-1</Value>
    <MetaElement REF="Example" />
  </IlisMeta07.ModelData.MetaAttribute>
</IlisMeta07.ModelData>
```

7 Exclusion de responsabilité – Droits de tiers

Les normes élaborées par l'Association eCH et mises gratuitement à la disposition des utilisateurs, ainsi que les normes de tiers adoptées, ont seulement valeur de recommandations. L'Association eCH ne peut en aucun cas être tenue pour responsable des décisions ou mesures prises par un utilisateur sur la base des documents qu'elle met à disposition. L'utilisa-

teur est tenu d'étudier attentivement les documents avant de les mettre en application et au besoin de procéder aux consultations appropriées. Les normes **eCH** ne remplacent en aucun cas les consultations techniques, organisationnelles ou juridiques appropriées dans un cas concret.

Les documents, méthodes, normes, procédés ou produits référencés dans les normes **eCH** peuvent le cas échéant être protégés par des dispositions légales sur les marques, les droits d'auteur ou les brevets. L'obtention des autorisations nécessaires auprès des personnes ou organisations détentrices des droits relève de la seule responsabilité de l'utilisateur.

Bien que l'Association **eCH** mette tout en œuvre pour assurer la qualité des normes qu'elle publie, elle ne peut fournir aucune assurance ou garantie quant à l'absence d'erreur, l'actualité, l'exhaustivité et l'exactitude des documents et informations mis à disposition. La teneur des normes **eCH** peut être modifiée à tout moment sans préavis.

Toute responsabilité relative à des dommages que l'utilisateur pourrait subir par suite de l'utilisation des normes **eCH** est exclue dans les limites des réglementations applicables.

8 Droits d'auteur

Tout auteur de normes **eCH** en conserve la propriété intellectuelle. Il s'engage toutefois à mettre gratuitement, et pour autant que ce soit possible, la propriété intellectuelle en question ou ses droits à une propriété intellectuelle de tiers à la disposition des groupes de spécialistes respectifs ainsi qu'à l'association **eCH**, pour une utilisation et un développement sans restriction dans le cadre des buts de l'association.

Les normes élaborées par les groupes de spécialistes peuvent, moyennant mention des auteurs **eCH** respectifs, être utilisées, développées et déployées gratuitement et sans restriction.

Les normes **eCH** sont complètement documentées et libres de toute restriction relevant du droit des brevets ou de droits de licence. La documentation correspondante peut être obtenue gratuitement.

Les présentes dispositions s'appliquent exclusivement aux normes élaborées par **eCH**, non aux normes ou produits de tiers auxquels il est fait référence dans les normes **eCH**. Les normes incluront les références appropriées aux droits de tiers.

Annexe– Références et Bibliographie

- [1] Groupe spécialisé eCH - Géoinformation (2006): *eCH-0031: Géoinformation: INTERLIS 2 – Manuel de référence*. Norme eCH
- [2] KOGIS (2008): *Méta-modèle -INTERLIS 2*.
Online http://www.interlis.ch/interlis2/docs23/ili2-metamodel_2008-08-29_d.pdf